Legendre Expansions & Spherical Harmonics

William O. Bray

Basic Plots

The following commands give plots similar to Figures 8.2 and 8.3 in the text.

```
p1 := Plot[LegendreP[n, s] /. n → {1, 3, 5, 7, 9},
  {s, -1, 1}, PlotLabel → "Odd Degree", ImageSize → {300, 200}]
p2 := Plot[LegendreP[n, s] /. n → {0, 2, 4, 6, 8}, {s, -1, 1},
  PlotLabel → "Even Degree", ImageSize → {300, 200}]
GraphicsRow[{p1, p2}]
```



As functions on the sphere, we can look at the plots with the following commands.

```
g[n_] := SphericalPlot3D[1 + LegendreP[n, Cos[t]], {t, 0, Pi},
  {tt, 0, 2 Pi}, Mesh → None, PlotPoints → 25, Boxed → False, Axes → False]
GraphicsGrid[{{g[2], g[4], g[6]}, {g[3], g[5], g[7]}}]
```



An Expansion Example (*Mathematica* version of Example 8.1 in text)

```
f[x_] := Piecewise[{{-1, -1 ≤ x < 0}, {1, 0 ≤ x < 1}}]
```

```
(2 l + 1) Integrate[f[x] LegendreP[l, x], {x, -1, 1}] / 2
```

$$\left((1+2\,l)\left(l\,\pi^{3/2}+l^2\,\pi^{3/2}-2\,\text{Gamma}\left[1-\frac{l}{2}\right]\,\text{Gamma}\left[\frac{3+l}{2}\right]\,\text{Sin}[l\,\pi]\right)\right)\Big/\left(2\,l\,(1+l)\,\pi\,\text{Gamma}\left[1-\frac{l}{2}\right]\,\text{Gamma}\left[\frac{3+l}{2}\right]\right)$$

```
Assuming[l ∈ Integers, Simplify[%]]
```

$$\frac{(1+2\,l)\,\sqrt{\pi}}{2\,\text{Gamma}\left[1-\frac{l}{2}\right]\,\text{Gamma}\left[\frac{3+l}{2}\right]}$$

```
fhat = Table[%, {l, 1, 25}]
```

$$\left\{\frac{3}{2},\ 0,\ -\frac{7}{8},\ 0,\ \frac{11}{16},\ 0,\ -\frac{75}{128},\ 0,\ \frac{133}{256},\ 0,\ -\frac{483}{1024},\ 0,\ \frac{891}{2048},\ 0,\ -\frac{13\,299}{32\,768},\ 0,\ \frac{25\,025}{65\,536},\ 0,\ -\frac{94\,809}{262\,144},\ 0,\ \frac{180\,557}{524\,288},\ 0,\ -\frac{1\,381\,471}{4\,194\,304},\ 0,\ \frac{2\,652\,153}{8\,388\,608}\right\}$$

Notice that the Legendre coefficient corresponding to $l = 0$ is zero.

```
ps[s_, N_] := Sum[fhat[[l]] LegendreP[l, s], {l, 1, N}]
```

```
Plot[{f[s], ps[s, 3], ps[s, 8], ps[s, 21]}, {s, -1, 1}]
```



Spherical Harmonics

The code below generates Figure 8.6 in the text. By varying the parameters other spherical harmonics can be plotted on the sphere.

```
splot[m_] :=
 SphericalPlot3D[1 + Re[SphericalHarmonicY[5, m, t, p]], {t, 0, Pi},
  {p, 0, 2 Pi}, Boxed → False, Mesh → None, PlotPoints → 25, Axes → False]
GraphicsGrid[{{splot[0], splot[1], splot[2]},
  {splot[3], splot[4], splot[5]}}]
```